

# Cmajor Programming Language Specification

## Version 3.3

Seppo Laakko

October 19, 2015

## Contents

<b>Contents</b>	<b>1</b>
<b>1 Grammar Notation</b>	<b>5</b>
<b>2 Lexical Structure</b>	<b>5</b>
2.1 White Space and Comments . . . . .	5
2.2 Keywords . . . . .	6
2.3 Identifiers . . . . .	6
2.4 Literals . . . . .	7
2.4.1 Boolean Literals . . . . .	7
2.4.2 Integer Literals . . . . .	7
2.4.3 Floating Literals . . . . .	7
2.4.4 Character Literals . . . . .	8
2.4.5 String Literals . . . . .	8
2.4.6 Null-literal . . . . .	8
<b>3 Basic Types</b>	<b>9</b>
3.1 Boolean Type . . . . .	9
3.2 Integer Types . . . . .	9
3.3 Floating-point Types . . . . .	10
3.4 Character Type . . . . .	10
3.5 Void Type . . . . .	10
<b>4 Expressions</b>	<b>10</b>
4.1 Equivalence . . . . .	11
4.2 Implication . . . . .	11
4.3 Disjunction . . . . .	11
4.4 Conjunction . . . . .	11
4.5 Bitwise OR-expression . . . . .	11
4.6 Bitwise XOR-expression . . . . .	11
4.7 Bitwise AND-expression . . . . .	12
4.8 Equality-expression . . . . .	12

4.9	Relational Expression . . . . .	12
4.9.1	Is and As Operators . . . . .	12
4.10	Shift Expression . . . . .	13
4.11	Additive Expression . . . . .	14
4.12	Multiplicative Expression . . . . .	14
4.13	Prefix Expression . . . . .	14
4.14	Postfix Expression . . . . .	15
4.14.1	Postfix Increment and Decrement Operators . . . . .	15
4.14.2	Member Access Operator . . . . .	15
4.14.3	Pointer Member Access Operator . . . . .	16
4.14.4	Invocation Operator . . . . .	16
4.14.5	Indexing Operator . . . . .	16
4.15	Primary Expression . . . . .	16
4.16	Size-of Expression . . . . .	16
4.17	Cast Expression . . . . .	17
4.18	New Expression . . . . .	17
4.19	Construct Expression . . . . .	17
4.20	Template Identifier . . . . .	17
4.21	Type Expression . . . . .	17
4.21.1	Prefix Type Expression . . . . .	17
4.21.2	Postfix Type Expression . . . . .	18
4.21.3	Primary Type Expression . . . . .	18
4.22	Constant Expression . . . . .	18
<b>5</b>	<b>Statements</b>	<b>18</b>
5.1	Control Statements . . . . .	19
5.1.1	Return Statement . . . . .	19
5.1.2	Conditional Statement . . . . .	19
5.1.3	Switch Statement . . . . .	19
5.1.4	While Statement . . . . .	20
5.1.5	Do Statement . . . . .	20
5.1.6	Range-for Statement . . . . .	20
5.1.7	For Statement . . . . .	20
5.1.8	Compound Statement . . . . .	20
5.1.9	Break Statement . . . . .	20
5.1.10	Continue Statement . . . . .	21
5.1.11	Goto Statement . . . . .	21
5.2	Typedef Statement . . . . .	21
5.3	Simple Statement . . . . .	21
5.4	Assignment Statement . . . . .	21
5.5	Construction Statement . . . . .	21
5.6	Delete Statement . . . . .	21
5.7	Destroy Statement . . . . .	22
5.8	Throw Statement . . . . .	22
5.9	Try-Catch Statement . . . . .	22
5.10	Assert Statement . . . . .	22
5.11	Conditional Compilation Statement . . . . .	22

<b>6</b>	<b>Access Specifiers</b>	<b>23</b>
<b>7</b>	<b>Functions</b>	<b>24</b>
7.1	Function Specifiers . . . . .	24
7.1.1	Inline Specifier . . . . .	24
7.1.2	CDecl Specifier . . . . .	25
7.1.3	Nothrow Specifier . . . . .	25
7.1.4	Throw Specifier . . . . .	25
7.1.5	Unit Test Specifier . . . . .	25
7.2	Function Templates . . . . .	25
7.2.1	Constrained Function Templates . . . . .	25
7.2.2	Function Template and Overload Resolution . . . . .	25
<b>8</b>	<b>Derived Types</b>	<b>25</b>
8.1	Pointer Types . . . . .	25
8.1.1	Generic Pointer Type . . . . .	26
8.1.2	Null Pointer Type . . . . .	26
8.2	Lvalue Reference Types . . . . .	26
8.3	Rvalue Reference Types . . . . .	26
8.4	Array Types . . . . .	26
8.5	Constant Types . . . . .	26
<b>9</b>	<b>User-defined Types</b>	<b>27</b>
9.1	Enumerated Types . . . . .	27
9.2	Class Types . . . . .	27
9.2.1	Regular Class Types . . . . .	27
9.2.2	Static Class Types . . . . .	28
9.2.3	Abstract Class Types . . . . .	28
9.2.4	Class Templates . . . . .	28
9.2.5	Inheritance . . . . .	28
9.2.6	Constrained Class Templates . . . . .	28
9.2.7	Static Constructor . . . . .	28
9.2.8	Constructors . . . . .	29
9.2.9	Destructor . . . . .	30
9.2.10	Member Functions . . . . .	31
9.2.11	Conversion Functions . . . . .	32
9.2.12	Member Variable Declarations . . . . .	32
9.2.13	Constant Declarations . . . . .	33
9.2.14	Typedef Declarations . . . . .	33
9.3	Delegate Types . . . . .	33
9.3.1	Delegates . . . . .	33
9.3.2	Class Delegates . . . . .	33
<b>10</b>	<b>Concepts</b>	<b>33</b>
10.1	Typename Constraint . . . . .	34
10.2	Signature Constraint . . . . .	34
10.3	Embedded Constraint . . . . .	35

10.4	Axioms . . . . .	35
10.5	Where Constraint . . . . .	35
10.5.1	Disjunctive Constraint Expressions . . . . .	35
10.5.2	Conjunctive Constraint Expressions . . . . .	35
10.5.3	Primary Constraint Expressions . . . . .	35
10.5.4	Atomic Constraints . . . . .	36
10.5.5	Is-Constraint . . . . .	36
10.5.6	Multiparameter Constraint . . . . .	36
10.6	Built-in Concepts . . . . .	36
10.6.1	Same Concept . . . . .	36
10.6.2	Derived Concept . . . . .	36
10.6.3	Convertible Concept . . . . .	36
10.6.4	Common Concept . . . . .	37
<b>11</b>	<b>Namespaces</b>	<b>37</b>
<b>12</b>	<b>Source Files</b>	<b>37</b>
<b>13</b>	<b>Programs</b>	<b>37</b>
<b>14</b>	<b>Solution and Project File Formats</b>	<b>38</b>
14.1	Solution File Format . . . . .	38
14.2	Project File Format . . . . .	38
	<b>References</b>	<b>40</b>

# 1 Grammar Notation

The lexical and syntactical structure of Cmajor programs is presented in this text using grammars that are in extended Backus-Naur form. A grammar consists of

1. *Terminal* symbols that are the elementary symbols of the language generated by the grammar. Terminals are presented like **this** in a type-writer font.
2. *Nonterminal* symbols also called syntactic variables that represent sets of strings of terminal symbols. Nonterminals are presented like *this* in italic font within angle brackets.
3. *Operators* ( ), |, −, \*, +, ?, [ ], and \ that operate on strings of terminals and nonterminals. If an operator character is ment to be a terminal symbol, it is quoted like this: ‘\*’. The meaning of each operator appears in table 1.
4. *Productions* each of which consists of a nonterminal symbol called the *head* of the production, symbol ::= (pronounced “produces”), and a sequence of terminals, nonterminals and operators collectively called the *body* of the production.

Table 1: Grammar Operators

Expression	Meaning
$\alpha \mid \beta$	$\alpha$ or $\beta$
$\alpha - \beta$	$\alpha$ but not $\beta$
$\alpha^*$	zero or more $\alpha$ ’s
$\alpha^+$	one or more $\alpha$ ’s
$\alpha^?$	zero or one $\alpha$ ’s
$(\alpha \beta)$	grouping of $\alpha$ and $\beta$ together
$[\mathbf{a-z}]$	a terminal character in range a ... z
$[\text{^}\backslash\mathbf{r}\backslash\mathbf{na-c}]$	any terminal character except a carriage return, a newline or a character in range a ... c

## 2 Lexical Structure

### 2.1 White Space and Comments

Lexical elements in program texts can be separated by one or more instance of white space characters and comments.

$\langle \text{spaces-and-comments} \rangle ::= (\langle \text{space} \rangle \mid \langle \text{comment} \rangle)^+$

$\langle \text{space} \rangle ::=$  ‘any ASCII character classified as space character according to C isspace function’

$\langle \text{comment} \rangle ::= \langle \text{line-comment} \rangle \mid \langle \text{block-comment} \rangle$

$\langle \text{line-comment} \rangle ::= // [\text{^}\backslash\mathbf{r}\backslash\mathbf{n}]^* \langle \text{new-line} \rangle$

$\langle \text{new-line} \rangle ::= \backslash \text{r} \backslash \text{n} \mid \backslash \text{n} \mid \backslash \text{r}$

$\langle \text{block-comment} \rangle ::= '/*' (\langle \text{any-char} \rangle - '*/') '*/'$

$\langle \text{any-char} \rangle ::= \text{'any 8-bit ASCII character'}$

## 2.2 Keywords

Keywords have reserved context-dependent meaning in a program and they cannot be used as identifiers.

$\langle \text{keyword} \rangle ::= \text{'see table 2'}$

Table 2: Keywords

<b>abstract</b>	<b>and</b>	<b>as</b>	<b>axiom</b>	<b>base</b>	<b>bool</b>
<b>break</b>	<b>byte</b>	<b>case</b>	<b>cast</b>	<b>catch</b>	<b>cdecl</b>
<b>char</b>	<b>class</b>	<b>concept</b>	<b>const</b>	<b>construct</b>	<b>continue</b>
<b>default</b>	<b>delegate</b>	<b>delete</b>	<b>destroy</b>	<b>do</b>	<b>double</b>
<b>else</b>	<b>enum</b>	<b>explicit</b>	<b>extern</b>	<b>false</b>	<b>float</b>
<b>for</b>	<b>goto</b>	<b>if</b>	<b>inline</b>	<b>int</b>	<b>internal</b>
<b>is</b>	<b>long</b>	<b>namespace</b>	<b>new</b>	<b>not</b>	<b>nothrow</b>
<b>null</b>	<b>operator</b>	<b>or</b>	<b>override</b>	<b>private</b>	<b>protected</b>
<b>public</b>	<b>return</b>	<b>sbyte</b>	<b>short</b>	<b>sizeof</b>	<b>static</b>
<b>suppress</b>	<b>switch</b>	<b>this</b>	<b>throw</b>	<b>true</b>	<b>try</b>
<b>typedef</b>	<b>typename</b>	<b>uint</b>	<b>ulong</b>	<b>unit_test</b>	<b>ushort</b>
<b>using</b>	<b>virtual</b>	<b>void</b>	<b>where</b>	<b>while</b>	

## 2.3 Identifiers

Identifiers are used to name entities in a program. Identifiers that begin with double underscores are reserved for the compiler.

$\langle \text{identifier} \rangle ::= \langle \text{id-char-sequence} \rangle - \langle \text{keyword} \rangle$  [2.2](#)

$\langle \text{id-char-sequence} \rangle ::= (\langle \text{letter} \rangle \mid \_)(\langle \text{letter} \rangle \mid \_ \mid \langle \text{digit} \rangle)^*$

$\langle \text{letter} \rangle ::= [\text{A-Za-z}]$

$\langle \text{digit} \rangle ::= [0-9]$

$\langle \text{qualified-id} \rangle ::= \langle \text{identifier} \rangle (\cdot \langle \text{identifier} \rangle)^*$

## 2.4 Literals

Literals are used to write values in a program.

$\langle \textit{literal} \rangle ::= \langle \textit{boolean-literal} \rangle$  [2.4.1](#)  
|  $\langle \textit{integer-literal} \rangle$  [2.4.2](#)  
|  $\langle \textit{floating-literal} \rangle$  [2.4.3](#)  
|  $\langle \textit{character-literal} \rangle$  [2.4.4](#)  
|  $\langle \textit{string-literal} \rangle$  [2.4.5](#)  
|  $\langle \textit{raw-string-literal} \rangle$  [2.4.5](#)  
|  $\langle \textit{null-literal} \rangle$  [2.4.6](#)

### 2.4.1 Boolean Literals

$\langle \textit{boolean-literal} \rangle ::= \text{true} \mid \text{false}$

The type of a boolean literal is **bool**.

### 2.4.2 Integer Literals

$\langle \textit{integer-literal} \rangle ::= \langle \textit{decimal-digit-sequence} \rangle \langle \textit{unsigned-suffix} \rangle?$   
|  $(\text{0x} \mid \text{0X}) \langle \textit{hexadecimal-digit-sequence} \rangle \langle \textit{unsigned-suffix} \rangle?$

$\langle \textit{decimal-digit-sequence} \rangle ::= \langle \textit{decimal-digit} \rangle^+$

$\langle \textit{decimal-digit} \rangle ::= [0-9]$

$\langle \textit{unsigned-suffix} \rangle ::= \text{'u'} \mid \text{'U'}$

$\langle \textit{hexadecimal-digit-sequence} \rangle ::= \langle \textit{hexadecimal-digit} \rangle^+$

$\langle \textit{hexadecimal-digit} \rangle ::= [0-9\text{a-fA-F}]$

If integer literal has no unsigned suffix, the type of the integer literal is first of the following types that can represent the value: **sbyte**, **byte**, **short**, **ushort**, **int**, **uint**, **long**, **ulong**.

If integer literal has an unsigned suffix, the type of the integer literal is first of the following types that can represent the value: **byte**, **ushort**, **uint**, **ulong**.

### 2.4.3 Floating Literals

$\langle \textit{floating-literal} \rangle ::= (\langle \textit{fractional-floating-literal} \rangle \mid \langle \textit{exponent-floating-literal} \rangle) \langle \textit{floating-suffix} \rangle?$

$\langle \textit{fractional-floating-literal} \rangle ::= \langle \textit{decimal-digit-sequence} \rangle? . \langle \textit{decimal-digit-sequence} \rangle \langle \textit{exponent-part} \rangle?$   
|  $\langle \textit{decimal-digit-sequence} \rangle .$

$\langle \textit{exponent-part} \rangle ::= (\text{e} \mid \text{E}) \langle \textit{sign} \rangle? \langle \textit{decimal-digit-sequence} \rangle$

$\langle \textit{sign} \rangle ::= + \mid -$

$\langle \textit{exponent-floating-literal} \rangle ::= \langle \textit{decimal-digit-sequence} \rangle \langle \textit{exponent-part} \rangle$

$\langle \text{floating-suffix} \rangle ::= \text{'f'} \mid \text{'F'}$

If a floating literal has a floating suffix, the type of it is **float**, otherwise the type of the floating literal is **double**.

#### 2.4.4 Character Literals

$\langle \text{character-literal} \rangle ::= \text{' } ([^{\backslash} \text{r} \backslash \text{n}] \mid \langle \text{escape} \rangle) \text{'}$

$\langle \text{escape} \rangle ::= \backslash (\langle \text{hex-escape} \rangle \mid \langle \text{decimal-escape} \rangle \mid \langle \text{char-escape} \rangle)$

$\langle \text{hex-escape} \rangle ::= [\text{xX}] \langle \text{hexadecimal-digit-sequence} \rangle$

$\langle \text{decimal-escape} \rangle ::= [\text{dD}] \langle \text{decimal-digit-sequence} \rangle$

$\langle \text{char-escape} \rangle ::= [\text{^dDxX} \backslash \text{r} \backslash \text{n}] \text{'see table 3'}$

Table 3: Character Escape Sequences

Escape Sequence	Character Name	Character Code
$\backslash \text{n}$	new line	10
$\backslash \text{t}$	horizontal tab	9
$\backslash \text{v}$	vertical tab	11
$\backslash \text{b}$	backspace	8
$\backslash \text{r}$	carriage return	13
$\backslash \text{f}$	form feed	12
$\backslash \text{a}$	alert	7
$\backslash \backslash$	backslash	92
$\backslash 0$	null	0
$\backslash \text{x}$	any other character x stands for itself	

The type of a character literals is **char**.

#### 2.4.5 String Literals

The backslash character provides an escaping mechanism for ordinary string literals.

$\langle \text{string-literal} \rangle ::= \text{" } ([^{\backslash} \text{"} \backslash \text{r} \backslash \text{n}] \mid \langle \text{escape} \rangle)^* \text{"}$

There is no escapes in raw string literals. The content is taken literally.

$\langle \text{raw-string-literal} \rangle ::= \text{@ " } ([^{\text{"}}])^* \text{"}$

The type of a string literal and raw string literal is **const char\***.

#### 2.4.6 Null-literal

$\langle \text{null-literal} \rangle ::= \text{null}$

The null literal can be implicitly converted to any pointer or delegate type.



### 3 Basic Types

Cmajor has basic types for representing Boolean, integer, floating-point and character values, and for expressing lack of type.

```
⟨basic-type⟩ ::= bool
| sbyte
| byte
| short
| ushort
| int
| uint
| long
| ulong
| float
| double
| char
| void
```

#### 3.1 Boolean Type

The Boolean type **bool** is an 8-bit integral type with a value 1 representing **true** and value 0 representing **false**.

Operators unary **!**, and binary **||**, **&&** and **==** operate on Boolean type operands.

The default value of a **bool** type object is **false**.

A Boolean type value can be converted to an integer type, floating-point type or **char** type with an explicit **cast**.

#### 3.2 Integer Types

**sbyte** is an 8-bit signed integer type for representing values in range  $-128 \dots 127$ .

**byte** is an 8-bit unsigned integer type for representing values in range  $0 \dots 255$ .

**short** is a 16-bit signed integer type for representing values in range  $-32768 \dots 32767$ .

**ushort** is a 16-bit unsigned integer type for representing values in range  $0 \dots 65535$ .

**int** is a 32-bit signed integer type for representing values in range  $-2147483648 \dots 2147483647$ .

**uint** is a 32-bit unsigned integer type for representing values in range  $0 \dots 4294967295$ .

**long** is a 64-bit signed integer type for representing values in range  $-9223372036854775808 \dots 9223372036854775807$ .

**ulong** is a 64-bit unsigned integer type for representing values in range  $0 \dots 18446744073709551615$ .

The default value of an integer type is 0.

Unary operators **-**, **+**, **~**, prefix operators **++** and **--**, binary operators **+**, **-**, **\***, **/**, **%**, **==**, **<**, **&**, **|**, **^**, **<<**, and **>>** operate on integer type operands.

There is an implicit conversion from a signed or unsigned integer type whose bit width is smaller to a signed integer type whose bit width is larger. For example, a **sbyte** can be implicitly converted to an **int**, because the bit width of **sbyte** is 8 and the bit width of **int** is 32. There is an implicit conversion from an unsigned integer type whose bit width is smaller to an unsigned integer type whose bit width is larger. For example, a **ushort** can be implicitly

converted to an **uint**, because the bit width of **ushort** is 16 and the bit width of **uint** is 32. Other conversions between integer types are explicit (they need a **cast**).

There is an implicit conversion from integer types whose bit width is less than or equal to 32 bits to **float** floating-point type.

There is an implicit conversion from all integer types to **double** floating-point type.

An integer type value can be converted to a **char** or **bool** value with an explicit **cast**.

### 3.3 Floating-point Types

Floating-point type **float** is a 32-bit IEEE 754 type. The default value of a **float** type object is 0.0f.

Floating-point type **double** is a 64-bit IEEE 754 type. The default value of a **double** type object is 0.0.

There is an implicit conversion from **float** to **double**.

A **double** type value can be converted to **float** with an explicit **cast**.

A floating-point type value can be converted to integer type, **char** type or **bool** type with an explicit **cast**.

Operators unary  $-$ ,  $+$ , binary  $+$ ,  $-$ ,  $*$ ,  $/$ ,  $==$  and  $<$  operate on floating-point type operands. A floating-point type value can be converted to an integer type with an explicit **cast** operator.

### 3.4 Character Type

The character type **char** is a 8-bit integral ASCII code type for representing character code values in range  $0 \dots 255$ . The default value of a **char** type object is `'\0'`.

Character code values can be compared for equality with operator  $==$  and for less-than relationship with operator  $<$ .

A **char** value can be converted to an integer type, floating-point type or **bool** type with an explicit **cast**.

### 3.5 Void Type

The type **void** represents lack of type. It is an incomplete type that cannot be completed, meaning you cannot have an object of type **void**.

## 4 Expressions

An expression represents a computation that usually produces a value <sup>1</sup>. This grammar accepts a superset of valid expressions. The implementation contains special disambiguation rules that reject syntactically valid (according to the grammar) but meaningless expressions. The type checking phase of the compiler further rejects semantically invalid expressions.

$\langle expression \rangle ::= \langle equivalence \rangle$ [4.1](#)

---

<sup>1</sup>Calling a void function does not produce a value.

## 4.1 Equivalence

$\langle equivalence \rangle ::= \langle implication \rangle_{4.2} (<=> \langle implication \rangle)^*$

The  $<=>$  operator is used in axioms (10.4) to state that operand expressions are logically equivalent. It groups operands from left to right.

## 4.2 Implication

$\langle implication \rangle ::= \langle disjunction \rangle_{4.3} (=> \langle implication \rangle)?$

The  $=>$  operator is used in axioms (10.4) to state that the right operand is logical implication of the left operand. It groups operands from right to left.

## 4.3 Disjunction

$\langle disjunction \rangle ::= \langle conjunction \rangle_{4.4} (|| \langle conjunction \rangle)^*$

The  $||$  operator takes boolean operands and yields a boolean value. It groups operands from left to right. Expression  $a||b$  is **true** when  $a$  is **true**, or  $a$  is **false** and  $b$  is **true** (in the first case operand  $b$  is not evaluated); otherwise **false**.

A user-defined class cannot overload the  $||$  operator.

## 4.4 Conjunction

$\langle conjunction \rangle ::= \langle bit-or-expr \rangle_{4.5} (\&\& \langle bit-or-expr \rangle)^*$

The  $\&\&$  operator takes boolean operands and yields a boolean value. It groups operands from left to right. Expression  $a\&\&b$  is **false** when  $a$  is **false**, or  $a$  is **true** and  $b$  is **false** (in the first case operand  $b$  is not evaluated); otherwise **true**.

A user-defined class cannot overload the  $\&\&$  operator.

## 4.5 Bitwise OR-expression

$\langle bit-or-expr \rangle ::= \langle bit-xor-expr \rangle_{4.6} (| \langle bit-xor-expr \rangle)^*$

The  $|$  operator with integer operands yields an integer value. It groups operands from left to right. The result is a bitwise OR of its operands: that is, each bit of the result will be 1 if one or both of the corresponding bits of the operands is 1; 0 otherwise.

A user-defined class can overload the  $|$  operator by implementing the **operator|** function.

## 4.6 Bitwise XOR-expression

$\langle bit-xor-expr \rangle ::= \langle bit-and-expr \rangle_{4.7} (^ \langle bit-and-expr \rangle)^*$

The  $\wedge$  operator with integer operands yields an integer value. It groups operands from left to right. The result is a bitwise XOR of its operands: that is, each bit of the result will be 1 if either but not both of the corresponding bits of the operands is 1; 0 otherwise.

A user-defined class can overload the  $\wedge$  operator by implementing the **operator^** function.

## 4.7 Bitwise AND-expression

$\langle \text{bit-and-expr} \rangle ::= \langle \text{equality-expr} \rangle_{4.8} (\& \langle \text{equality-expr} \rangle)^*$

The `&` operator with integer operands yields an integer value. It groups operands from left to right. The result is a bitwise AND of its operands: that is, each bit of the result will be 1 if both of the corresponding bits of the operands is 1; 0 otherwise.

A user-defined class can overload the `&` operator by implementing the **operator&** function.

## 4.8 Equality-expression

$\langle \text{equality-expr} \rangle ::= \langle \text{relational-expr} \rangle_{4.9} ((== \mid !=) \langle \text{relational-expr} \rangle)^*$

The `==` operator compares equality. It groups operands from left to right.

The compiler will automatically implement the `!=` operator for a type if it implements the `==` operator. Then expression `a != b` will be evaluated as **operator!(operator==(a,b))**.

A user-defined class can overload the `==` and `!=` operators by implementing the **operator==** function.

## 4.9 Relational Expression

$\langle \text{relational-expr} \rangle ::= \langle \text{shift-expr} \rangle_{4.10} ((< \mid > \mid <= \mid >=) \langle \text{shift-expr} \rangle)^* \\ \mid \langle \text{shift-expr} \rangle ((\text{is} \mid \text{as}) \langle \text{type-expr} \rangle_{4.21})^*$

The `<` operator compares less-than relationship. It groups operands from left to right.

The compiler will automatically implement operators `>`, `<=` and `>=` for a type if it implements the `<` operator. Then expression `a > b` will be evaluated as **operator<(b,a)**, expression `a <= b` as **operator!(operator<(b,a))**, expression `a >= b` as **operator!(operator<(a,b))**.

A user-defined class can overload the `<` and `>`, `<=` and `>=` operators by implementing the **operator<** function.

### 4.9.1 Is and As Operators

Consider following code:

```
1 public class Base
2 {
3     public virtual ~Base()
4     {
5     }
6     // ...
7 }
8
9 public class Derived : Base
10 {
11     // ...
12 }
13
14 public Base* GetBasePtrFromSomewhere()
15 {
```

```

16     return new Derived();
17 }
18
19 void main()
20 {
21     Base* b1 = GetBasePtrFromSomewhere();
22     if (b1 is Derived*)
23     {
24         // do something with b1
25     }
26
27     Base* b2 = GetBasePtrFromSomewhere();
28     Derived* d = b2 as Derived*;
29     if (d != null)
30     {
31         // do something with d
32     }
33 }

```

The **is** operator tests if the left operand can be legally casted to a pointer type on the right-hand side. That is: if the pointer *b1* in fact points to a *Derived* class object or an object of class derived from *Derived*. The left operand of the **is** operator must be a pointer to a virtual class object<sup>2</sup> and the right operand must be a pointer to virtual class type. The **is** operator yields a Boolean result.

The **as** operator tries to convert the left operand to a pointer type on the right-hand side. If the conversion succeeds you got a non-null pointer to *Derived* class, otherwise the result is **null**. In the case of previous example the conversion succeeds if pointer *b2* in fact points to *Derived* class object or an object of class derived from *Derived*. The left operand of the **as** operator must be pointer to a virtual class object and the right operand must be a pointer to virtual class type. The **as** operator yields **null** or non-null pointer result. It performs an operation similar to C++ **dynamic\_cast**.

Note: in the **debug**, **release** and **profile** configurations the **is** and **as** operators generate a call to a function that traverses the class hierarchy, but in **full** configuration the class identifiers are chosen so that the **is** and **as** generate only a single modulo operation in addition to retrieving the class identifier from the run-time type information table. In the **full** configuration the compiler does whole-program analysis and implements a scheme described in [http://www.stroustrup.com/fast\\_dynamic\\_casting.pdf](http://www.stroustrup.com/fast_dynamic_casting.pdf) by Michael Gibbs and Bjarne Stroustrup.

## 4.10 Shift Expression

$\langle \text{shift-expr} \rangle ::= \langle \text{additive-expr} \rangle 4.11 ((\ll | \gg) \langle \text{additive-expr} \rangle)^*$

The  $\ll$  operator with integer operands will yield an integer value. It groups operands from left to right. Expression  $a \ll b$  with integer operands  $a$  and  $b$  will yield a value of  $a$  shifted  $b$  bit positions left and filling the vacant bit positions of  $a$  with 0-bits from right. If  $a$  is non-negative and  $b$  small enough the result will be  $2^b a$ .

<sup>2</sup>to a class object containing virtual, abstract or overridden member functions

The `>>` operator with integer operands will yield an integer value. It groups operands from left to right. Expression  $a \gg b$  with integer operands  $a$  and  $b$  will yield a value of  $a$  shifted  $b$  bit positions right and filling the vacant bit positions of  $a$  with 0-bits from left. If  $a$  is non-negative the result will be  $a/2^b$  ('/' meaning integer division).

A user-defined class can overload the `<<` and `>>` operators by implementing the **operator<<** and **operator>>** functions respectively.

#### 4.11 Additive Expression

$\langle \text{additive-expr} \rangle ::= \langle \text{multiplicative-expr} \rangle$ [4.12](#)  $((+ \mid -) \langle \text{multiplicative-expr} \rangle)^*$

The `+` and `-` operators with arithmetic type operands will yield same arithmetic type values. If one of the operands is integer and one is floating-point type the integer type operand will be first converted to a floating-point type. Resulting values with operands  $a$  and  $b$  will be  $a + b$  and  $a - b$  respectively, if possible overflow or loss of precision is not taken account.

For `+` and `-` operators it is also possible that the left operand is of a pointer type ( $p$ ) and the right operand is of integer type ( $i$ ). In that case the expression  $p + i$  yields a pointer value pointing  $i$  objects after  $p$  in memory and expression  $p - i$  yields a pointer value pointing  $i$  objects before  $p$  in memory.

For `-` operator it is also possible that both operands are of a pointer type ( $p$  and  $q$ ). In that case the expression  $p - q$  yields an integer counting the number of objects between pointers  $p$  and  $q$ .

A user-defined class can overload the `+` and `-` operators by implement the **operator+** and **operator-** functions respectively.

#### 4.12 Multiplicative Expression

$\langle \text{multiplicative-expr} \rangle ::= \langle \text{prefix-expr} \rangle$ [4.13](#)  $((\ast \mid / \mid \%) \langle \text{prefix-expr} \rangle)^*$

The `*` and `/` operators with arithmetic type operands will yield same arithmetic type values. They groups operands from left to right. If one of the operands is integer and one is floating-point type the integer type operand will be first converted to a floating-point type. Resulting values with operands  $a$  and  $b$  will be  $ab$  and  $a/b$  respectively, if possible overflow or loss of precision is not taken account.

The `%` operator with integer operands will yield integer type values. It groups operands from left to right. The resulting value will be remainder of integer division  $a/b$ .

A user-defined class can overload the `*`, `/` and `%` operators by implement the **operator\***, **operator/**, and **operator%** functions respectively.

#### 4.13 Prefix Expression

$\langle \text{prefix-expr} \rangle ::= ( ++ \mid -- \mid + \mid - \mid ! \mid \sim \mid \& \mid \ast ) \langle \text{prefix-expr} \rangle \mid \langle \text{postfix-expr} \rangle$ [4.14](#)

The prefix `++` and `--` operators can have an integer or pointer type operand. The operand shall be an lvalue<sup>3</sup>. When the operand is of integer type, the `++` operator will increment its operand by 1 and the operator `--` will decrement its operand by 1. When the operand is of pointer type with pointer  $p$  pointing to an object of type  $T$ , the expression `++p`

---

<sup>3</sup>A variable, a reference or a result of a dereference operator.

increments pointer  $p$  so that  $p$  points to next object of type  $T$  in memory. Correspondingly  $--p$  decrements pointer  $p$  so that  $p$  points to previous object of type  $T$  in memory. The result of an increment or decrement operator is a value of the same type as the operand after incrementing or decrementing it.

The prefix  $+$  and  $-$  operators can have an arithmetic type operand. Expression  $+p$  will yield  $p$  and  $-p$  will yield the negation of  $p$ .

The prefix  $!$  operator can have a boolean operand. The result of  $!\text{true}$  is **false** and  $!\text{false}$  is **true**.

The  $\sim$  operator can have an integer operand. It will yield a bitwise complement of its operand.

The  $\&$  operator shall have an lvalue operand. If the operand is an object of type  $T$ , it returns an address or memory location of the object. The resulting value will be a pointer pointing to the object and it will have a type  $T^*$ .

The  $*$  operator can have a pointer operand. If the operand  $p$  is of a type pointer to  $T$ , the expression  $*p$  dereferences  $p$  yielding an lvalue of type  $T$ .

A user-defined class can overload any prefix operator  $op$  by implementing the **operator** $op$  function.

## 4.14 Postfix Expression

$\langle \text{postfix-expr} \rangle ::= \langle \text{primary-expr} \rangle$  4.15 ( $++$  4.14.1 |  $--$  4.14.1  
|  $\cdot$  4.14.2  $\langle \text{identifier} \rangle$  2.3  
|  $\rightarrow$  4.14.3  $\langle \text{identifier} \rangle$   
|  $\langle \text{'('} \langle \text{argument-list} \rangle \text{'})'}$  4.14.4  
|  $\langle \text{'['} \langle \text{expression} \rangle$  4  $\langle \text{']'}$  4.14.5  $\rangle^*$

$\langle \text{argument-list} \rangle ::= \langle \text{expression-list} \rangle?$

$\langle \text{expression-list} \rangle ::= \langle \text{expression} \rangle (, \langle \text{expression} \rangle)^*$

### 4.14.1 Postfix Increment and Decrement Operators

The postfix  $++$  and  $--$  operators can have an integer or pointer type operand. The operand shall be an lvalue. The result of the operator is the value of the operand before incrementing or decrementing it. When the operand is of integer type, the  $++$  operator will increment its operand by 1 and the operator  $--$  will decrement its operand by 1. When the operand is of pointer type with pointer  $p$  pointing to an object of type  $T$ , the expression  $p++$  increments pointer  $p$  so that  $p$  points to next object of type  $T$  in memory. Correspondingly  $p--$  decrements pointer  $p$  so that  $p$  points to previous object of type  $T$  in memory.

A user-defined class cannot overload the postfix forms of  $++$  or  $--$  operators. They are automatically implemented by the compiler if the prefix versions of those operators are implemented.

### 4.14.2 Member Access Operator

The member access operator  $\cdot$  will access the member named by the right operand from the entity specified by the left operand. The left operand can be a namespace, a class type, or an enumerated type. The type of the expression is the type of the right operand.

#### 4.14.3 Pointer Member Access Operator

The pointer member access operator `->` will access a member named by the right operand from the entity specified by the left operand. The left operand can be a pointer type or class type object. In the latter case the class type shall implement the **operator**`->` function that returns a pointer or another class type object. The type of the expression is the type of the right operand.

#### 4.14.4 Invocation Operator

The invocation operator `()` will invoke a function specified by the left operand with the specified list of arguments. The left operand can be a function, a member, an object of a class type that implements the **operator**`()` function or a type. The type of the expression is the return type of the function.

#### 4.14.5 Indexing Operator

The left operand of the indexing operator `[]` shall be of a pointer type, an array type or a class type object. When the left operand is pointer  $p$ , the expression  $p[i]$  is evaluated as  $*(p + i)$ . When the left operand is an array, the expression  $a[i]$  returns a reference to the  $i$ 'th element of array  $a$ , where  $i$  can range 0 to  $N - 1$ , where  $N$  is the dimension of the array  $a$  (see 4.21.2). When the left operand is of a class type, it shall implement the **operator**`[]` function taking one parameter.

### 4.15 Primary Expression

```
<primary-expr> ::= '(' <expression> 4 ')'  
| <literal> 2.4  
| <basic-type> 3  
| <identifier> 2.3  
| typename '(' <expression> ')'  
| this  
| base  
| <sizeof-expr> 4.16  
| <cast-expr> 4.17  
| <new-expr> 4.18  
| <construct-expr> 4.19  
| <template-id> 4.20
```

An expression in parenthesis, a literal (2.4), a name of a basic type (3) and an identifier are primary expressions.

Keyword **typename** followed by an expression in parenthesis yields the full name of the dynamic type of the expression. The type of it is **const char\***.

Keyword **this** represents a pointer to the current class object in class member functions.

Keyword **base** represents a pointer to the base class object in class member functions.

#### 4.16 Size-of Expression

```
<sizeof-expr> ::= sizeof '(' <expression> 4 ')'
```



The operand of the **sizeof** operator can be a type or an entity having a type. If the type in question is T, the operator yields constant of type **ulong** that is equal to the size of object of type T in bytes.

#### 4.17 Cast Expression

$\langle \text{cast-expr} \rangle ::= \text{cast} < \langle \text{type-expression} \rangle_{4.21} > '(\langle \text{expression} \rangle_4)'$

The **cast** operator performs explicit type conversion. In order to the cast operation to succeed, there shall be a built-in conversion, a conversion function(9.2.11) in the source type, or a converting constructor in the target type taking one parameter of type of the source expression.

#### 4.18 New Expression

$\langle \text{new-expr} \rangle ::= \text{new} \langle \text{type-expression} \rangle_{4.21} '(\langle \text{argument-list} \rangle_{4.14})'$

The **new** operator reserves memory from the free store for an object of the requested type, constructs the object by issuing the **construct** operator with the specified arguments, and returns a pointer to the newly constructed object.

#### 4.19 Construct Expression

$\langle \text{construct-expr} \rangle ::= \text{construct} < \langle \text{type-expression} \rangle > \\ '(\langle \text{expression} \rangle_4 (, \langle \text{expression-list} \rangle_{4.14})? )'$

The **construct** operator takes a type, a generic pointer to memory and constructor arguments. It constructs an object of requested type to the memory by calling the constructor with arguments, and returns a pointer to the newly constructed object.

#### 4.20 Template Identifier

$\langle \text{template-id} \rangle ::= \langle \text{function-or-class-name} \rangle < \langle \text{type-expression} \rangle_{4.21} (, \langle \text{type-expression} \rangle)^* > \\ \langle \text{function-or-class-name} \rangle ::= \langle \text{qualified-id} \rangle_{2.3}$

A template identifier is a primary expression.

#### 4.21 Type Expression

$\langle \text{type-expression} \rangle ::= \langle \text{prefix-type-expr} \rangle$

##### 4.21.1 Prefix Type Expression

$\langle \text{prefix-type-expr} \rangle ::= \text{const} \langle \text{postfix-type-expr} \rangle \\ | \langle \text{postfix-type-expr} \rangle$

A prefix type expression is either a **const** qualified postfix type expression, or sole postfix type expression.

#### 4.21.2 Postfix Type Expression

$\langle \text{postfix-type-expr} \rangle ::= \langle \text{primary-type-expr} \rangle ( \cdot \cdot \langle \text{identifier} \rangle$ [2.3](#)  
|  $\&\&$   
|  $\&$   
|  $*$   
|  $[ \langle \text{array-dimension} \rangle ] )^*$

$\langle \text{array-dimension} \rangle ::= \langle \text{constant-expression} \rangle$ [4.22](#)

A postfix type expression is a primary type expression optionally followed by sequence of the following: a member selection operator `·` and an identifier, an rvalue reference type operator (see [8.3](#)), an lvalue reference type operator (see [8.2](#)), a pointer type operator (see [8.1](#)), or an array type operator (see [8.4](#)).

#### 4.21.3 Primary Type Expression

$\langle \text{primary-type-expr} \rangle ::= \langle \text{basic-type} \rangle$ [3](#)  
|  $\langle \text{template-id} \rangle$ [4.20](#)  
|  $\langle \text{identifier} \rangle$   
|  $'( \langle \text{prefix-type-expr} \rangle )'$

A primary type expression consists of basic types, template identifiers, identifiers and parenthesized prefix type expressions.

#### 4.22 Constant Expression

$\langle \text{constant-expression} \rangle ::= \langle \text{expression} \rangle$ [4](#)

A constant expression is an expression whose value can be obtained at compile time. A constant expression can contain literals, constants, enumeration constants and operators that do not involve taking an address of an object.

### 5 Statements

Statements control the program execution logic, declare type aliases, evaluate expressions, assign values to variables, or construct or destroy them. A statement can be labeled by an identifier.

$\langle \text{statement} \rangle ::= ( \langle \text{identifier} \rangle : )? ( \langle \text{control-statement} \rangle$ [5.1](#)  
|  $\langle \text{typedef-statement} \rangle$ [5.2](#)  
|  $\langle \text{simple-statement} \rangle$ [5.3](#)  
|  $\langle \text{assignment-statement} \rangle$ [5.4](#)  
|  $\langle \text{construction-statement} \rangle$ [5.5](#)  
|  $\langle \text{delete-statement} \rangle$ [5.6](#)  
|  $\langle \text{destroy-statement} \rangle$ [5.7](#)  
|  $\langle \text{throw-statement} \rangle$ [5.8](#)  
|  $\langle \text{try-catch-statement} \rangle$ [5.9](#)  
|  $\langle \text{assert-statement} \rangle$ [5.10](#)  
|  $\langle \text{conditional-compilation-statement} \rangle )$ [5.11](#)

## 5.1 Control Statements

The control statements control statement execution order.

$\langle \text{control-statement} \rangle ::= \langle \text{return-statement} \rangle$  5.1.1  
|  $\langle \text{conditional-statement} \rangle$  5.1.2  
|  $\langle \text{switch-statement} \rangle$  5.1.3  
|  $\langle \text{while-statement} \rangle$  5.1.4  
|  $\langle \text{do-statement} \rangle$  5.1.5  
|  $\langle \text{range-for-statement} \rangle$  5.1.6  
|  $\langle \text{for-statement} \rangle$  5.1.7  
|  $\langle \text{compound-statement} \rangle$  5.1.8  
|  $\langle \text{break-statement} \rangle$  5.1.9  
|  $\langle \text{continue-statement} \rangle$  5.1.10  
|  $\langle \text{goto-statement} \rangle$  5.1.11

### 5.1.1 Return Statement

The **return** statement returns control from a function to its caller. When the return type of the function is other than **void** the return statement is mandatory, and there shall be a return value expression, otherwise there shall be no return value expression, and the return statement is optional.

$\langle \text{return-statement} \rangle ::= \text{return } \langle \text{expression} \rangle$  4? ;

### 5.1.2 Conditional Statement

The conditional statement executes a statement if its condition expression is **true**. Otherwise, if there is an else statement, it is executed. The condition shall be a Boolean expression.

$\langle \text{conditional-statement} \rangle ::= \text{if } '(' \langle \text{expression} \rangle$  4  $)'$   $\langle \text{statement} \rangle$  5 (else  $\langle \text{statement} \rangle$ )?

### 5.1.3 Switch Statement

The **switch** statement evaluates its condition expression and executes the matching case statement. Each control path of the **case** statement shall end with **break**, **goto case** or **goto default** statement, that is: the control is not allowed to “fall through”. Multiple cases can be grouped into one **case** statement, though. The condition expression and each case expression shall evaluate to a compile time constant.

$\langle \text{switch-statement} \rangle ::= \text{switch } '(' \langle \text{constant-expression} \rangle$  4.22  $)'$   
 $\{ \langle \text{case-statement} \rangle \mid \langle \text{default-statement} \rangle \}^* \}$

$\langle \text{case-statement} \rangle ::= (\text{case } \langle \text{constant-expression} \rangle :)+$   
 $((\langle \text{identifier} \rangle :)? (\langle \text{goto-case-statement} \rangle \mid \langle \text{goto-default-statement} \rangle) \mid \langle \text{statement} \rangle)$  5)\*

$\langle \text{goto-case-statement} \rangle ::= \text{goto case } \langle \text{constant-expression} \rangle ;$

$\langle \text{goto-default-statement} \rangle ::= \text{goto default} ;$

$\langle \text{default-statement} \rangle ::= \text{default} : ((\langle \text{identifier} \rangle :)? \langle \text{goto-case-statement} \rangle) \mid \langle \text{statement} \rangle)^*$

#### 5.1.4 While Statement

The **while** statement repeatedly evaluates its condition expression and executes a statement as long as the condition evaluates to **true**. The condition shall be a Boolean expression.

$\langle \text{while-statement} \rangle ::= \text{while } '(' \langle \text{expression} \rangle_4 ') ' \langle \text{statement} \rangle_5$

#### 5.1.5 Do Statement

The **do** statement repeatedly executes a statement and then evaluates its condition expression until the condition evaluates to **false**. The condition shall be a Boolean expression.

$\langle \text{do-statement} \rangle ::= \text{do } \langle \text{statement} \rangle_5 \text{ while } '(' \langle \text{expression} \rangle_4 ') ' ;$

#### 5.1.6 Range-for Statement

The **range-for** statement iterates through a container. For each iteration, a variable is bound to each value in the container and then a statement is executed. The *container* shall be an expression that will yield an object that is a container having *Begin()* and *End()* member functions that return iterators to the beginning and one-past-the-end of a container respectively.

$\langle \text{range-for-statement} \rangle ::= \text{for } '(' \langle \text{type-expression} \rangle_{4.21} \langle \text{identifier} \rangle_{2.3} ':' \langle \text{container} \rangle ') ' \langle \text{statement} \rangle_5$

$\langle \text{container} \rangle ::= \langle \text{expression} \rangle_4$

#### 5.1.7 For Statement

The **for** statement first initializes a variable and then repeatedly evaluates a condition expression and executes a statement as long as the condition evaluates to **true**. After each execution cycle an expression that typically increments a condition variable is evaluated. The condition shall be a Boolean expression. The initialization statement is optional as are the condition and increment expressions in which case they are evaluated as **true**.

$\langle \text{for-statement} \rangle ::= \text{for } '(' \langle \text{for-init-statement} \rangle \langle \text{expression} \rangle_4 ? ; \langle \text{expression} \rangle ? ') ' \langle \text{statement} \rangle_5$

$\langle \text{for-init-statement} \rangle ::= \langle \text{assignment-statement} \rangle_{5.4} \mid \langle \text{construction-statement} \rangle_{5.5} \mid ;$

#### 5.1.8 Compound Statement

The compound statement executes a sequence of statements in order. The statement sequence can be empty.

$\langle \text{compound-statement} \rangle ::= \{ ' ' \langle \text{statement} \rangle_5^* ' ' \}$

#### 5.1.9 Break Statement

The **break** statement transfers control to the statement coming after its closest containing **switch**, **while**, **do** or **for** statement, if there is one; otherwise to the end of the function.

$\langle \text{break-statement} \rangle ::= \text{break} ;$

### 5.1.10 Continue Statement

The **continue** statement transfers control to the end of the closest containing **while**, **do** or **for** loop.

$\langle \text{continue-statement} \rangle ::= \text{continue} ;$

### 5.1.11 Goto Statement

The **goto** statement transfers control to the statement labeled with the matching identifier.

$\langle \text{goto-statement} \rangle ::= \text{goto } \langle \text{identifier} \rangle_{2.3} ;$

## 5.2 Typedef Statement

The **typedef** statement declares an alias name to a type expression.

$\langle \text{typedef-statement} \rangle ::= \text{typedef } \langle \text{type-expression} \rangle_{4.21} \langle \text{identifier} \rangle_{2.3} ;$

## 5.3 Simple Statement

The simple statement evaluates an expression if there is one, otherwise it does nothing. Typically the expression is a function call, or incrementation or decrementation of a variable.

$\langle \text{simple-statement} \rangle ::= \langle \text{expression} \rangle_{4?} ;$

## 5.4 Assignment Statement

The assignment statement assign a value to an target object. The target expression shall be an lvalue expression.

$\langle \text{assignment-statement} \rangle ::= \langle \text{expression} \rangle_{4} = \langle \text{expression} \rangle ;$

## 5.5 Construction Statement

The construction statement constructs a local variable. If the variable is not of a class type, it fits into a register, its address is not taken, and it is not passed by reference argument, a register for the variable is allocated, otherwise space for the variable is reserved from the execution stack. If an initialization expression or expression list is given, the constructor for the variable is called with the given arguments, otherwise the variable is default-constructed.

$\langle \text{construction-statement} \rangle ::= \langle \text{type-expression} \rangle_{4.21} \langle \text{identifier} \rangle_{2.3} \langle \text{initialization} \rangle? ;$

$\langle \text{initialization} \rangle ::= '(\langle \text{expression-list} \rangle_{4.14})' \mid = \langle \text{expression} \rangle_{4}$

## 5.6 Delete Statement

The **delete** statement calls the destructor of an object if there is one, and then releases the memory reserved for the object back to the free store. The expression shall evaluate to a pointer to the object. If the object has a virtual destructor, it is called, otherwise the destructor for the pointee type of the pointer type is called.

$\langle \text{delete-statement} \rangle ::= \text{delete } \langle \text{expression} \rangle_{4} ;$

## 5.7 Destroy Statement

The **destroy** statement calls the destructor for an object but does not release memory reserved for the object. The expression shall evaluate to a pointer to the object. If the object has a virtual destructor, it is called, otherwise the destructor for the pointee type of the pointer type is called.

$\langle \text{destroy-statement} \rangle ::= \text{destroy } \langle \text{expression} \rangle$ [4](#) ;

## 5.8 Throw Statement

The **throw** statement causes an exception object to be thrown. The expression shall construct a class object to throw. The exception object includes the source line number and the source file name of the throw statement.

$\langle \text{throw-statement} \rangle ::= \text{throw } \langle \text{expression} \rangle$ [4](#) ;

## 5.9 Try-Catch Statement

The **try-catch** statement executes statements in a try-block and if an exception is thrown, control is transferred to a matching exception handler. Each exception handler handles exceptions that are the same type as or a type derived from the type of the variable of the exception handler. First matching handler is executed, so the handlers should be ordered from the most specific to the most general.

$\langle \text{try-catch-statement} \rangle ::= \text{try } \langle \text{try-block} \rangle \langle \text{exception-handler} \rangle +$

$\langle \text{try-block} \rangle ::= \langle \text{compound-statement} \rangle$ [5.1.8](#)

$\langle \text{exception-handler} \rangle ::= \text{catch } '(' \langle \text{type-expression} \rangle$ [4.21](#)  $\langle \text{identifier} \rangle$ [2.3](#)  $)' \langle \text{catch-block} \rangle$

$\langle \text{catch-block} \rangle ::= \langle \text{compound-statement} \rangle$

## 5.10 Assert Statement

The **# assert** statement tests a condition that should always be true. If the condition is false, an error message containing the failed expression, a source line number and a source file name is issued and the program is exited.

Assert statements are skipped when compiling a program using **release** configuration.

$\langle \text{assert-statement} \rangle ::= \text{'\#'} \text{ assert } '(' \langle \text{expression} \rangle$ [4](#)  $)' \text{'\;'}$

## 5.11 Conditional Compilation Statement

The conditional compilation statement makes it possible to include statements in compilation based on conditional compilation symbols.

When compiling a program using **debug** configuration (default), the symbol `DEBUG` is defined. When compiling a program using **release** configuration, the symbol `RELEASE` is defined. When compiling a program using Cmajor compiler in Windows, a symbol `WINDOWS` is defined. When compiling a program using Cmajor compiler in Linux, a symbol `LINUX` is defined.

Users can define other conditional compilation symbols in IDE or using the -D command line option.

```

<conditional-compilation-statement> ::= <cond-comp-if-statement>

<cond-comp-if-statement> ::= '#' if '(' cond-comp-expr ')' <statement-list>
    ('#' elif '(' cond-comp-expr ')' <statement-list>)*
    ('#' else <statement-list>)? '#' endif

<cond-comp-expr> ::= <cond-comp-disjunction>

<cond-comp-disjunction> ::= <cond-comp-conjunction> (|| <cond-comp-conjunction>)*

<cond-comp-conjunction> ::= <cond-comp-prefix> (&& <cond-comp-prefix>)*

<cond-comp-prefix> ::= <cond-comp-not> | <cond-comp-primary>

<cond-comp-not> ::= '!' <cond-comp-prefix>

<cond-comp-primary> ::= <identifier>2.3 | '(' <cond-comp-expr> ')

<statement-list> ::= <statement>5*

```

The **cond-comp-if** statement includes a list of statements in compilation if its conditional compilation expression evaluates true, otherwise, if one of the **elif** conditional compilation expressions evaluates true, the statements in that **elif**-part is included in compilation, otherwise, if the **else**-part is given, the list of statements in **else**-part is included to compilation.

Conditional compilation disjunction expression evaluates to true, if one of conditional compilation conjunction expressions it contains evaluates to true.

Conditional compilation conjunction expression evaluates to true, if all conditional compilation prefix expressions it contains evaluate to true.

Conditional compilation prefix expression can be either conditional compilation not-expression or conditional compilation primary expression.

Conditional compilation not-expression evaluates to true if its conditional compilation prefix expression evaluates to false.

Conditional compilation primary expression can be an conditional compilation identifier or parenthesized conditional compilation expression.

Conditional compilation identifier evaluates to true if it is defined.

## 6 Access Specifiers

Most entities defined within a class or a namespace can be given access specifiers that grant or reject access for another entity.

- **public** access grants access from everywhere.
- **protected** access specifier grants access for an entity defined within the same or derived class and rejects it for other entities.

- **private** access specifier grants access for an entity defined within the same class and rejects it for other entities.
- **internal** access specifier grants access for an entity defined in the same program or library and rejects it for other entities.

$\langle \text{access-specifiers} \rangle ::= \langle \text{access-specifier} \rangle^*$

$\langle \text{access-specifier} \rangle ::= \text{public} \mid \text{protected} \mid \text{private} \mid \text{internal}$

If no access specifiers are given, the default access for a namespace level entity is **internal** access, and for a class level entity it is **private** access.

## 7 Functions

A function represents a unit of computation that can be executed from the other parts of the program or from the function body itself. When the function is called, each of the argument expressions of the function call is evaluated and bound to the corresponding parameter of the function. Then the function body is executed and after that the control returns to the caller.

$\langle \text{function-definition} \rangle ::= \langle \text{function-specifiers} \rangle \langle \text{type-expression} \rangle \text{4.21} \langle \text{function-name} \rangle$   
 $\langle \text{template-parameter-list} \rangle? \langle \text{parameters} \rangle \langle \text{where-constraint} \rangle \text{10.5?} \langle \text{function-body} \rangle$

$\langle \text{function-specifiers} \rangle ::= (\langle \text{access-specifier} \rangle \text{6} \mid \text{inline} \mid \text{cdecl} \mid \text{nothrow} \mid \text{throw} \mid \text{unit\_test})^*$

$\langle \text{function-name} \rangle ::= \langle \text{identifier} \rangle \text{2.3} \mid \langle \text{operator-function-name} \rangle$

$\langle \text{operator-function-name} \rangle ::= \text{operator} (\ll \mid \gg \mid == \mid < \mid = \mid ++ \mid -- \mid -> \mid + \mid '-' \mid * \mid / \mid \% \mid$   
 $\& \mid '|' \mid ! \mid \sim \mid [] \mid ())$

$\langle \text{template-parameter-list} \rangle ::= < \langle \text{template-parameter} \rangle (, \langle \text{template-parameter} \rangle)^* >$

$\langle \text{template-parameter} \rangle ::= \langle \text{identifier} \rangle \langle \text{default-template-parameter} \rangle?$

$\langle \text{default-template-parameter} \rangle ::= '=' \langle \text{type-expression} \rangle$

$\langle \text{parameters} \rangle ::= '(' \langle \text{parameter-list} \rangle? ')'$

$\langle \text{parameter-list} \rangle ::= \langle \text{parameter} \rangle (, \langle \text{parameter} \rangle)^*$

$\langle \text{parameter} \rangle ::= \langle \text{type-expression} \rangle \langle \text{identifier} \rangle?$

$\langle \text{function-body} \rangle ::= \langle \text{compound-statement} \rangle \text{5.1.8}$

### 7.1 Function Specifiers

#### 7.1.1 Inline Specifier

**inline** specifier is a hint to optimizer that it should replace function body in place of function call when applicable.



### 7.1.2 CDecl Specifier

**cdecl** specifier suppresses name mangling <sup>4</sup>, so that the function can easily be called from C language.

### 7.1.3 Nothrow Specifier

**nothrow** specifier states that the function will not throw Cmajor exceptions directly or indirectly, or that the function handles all exceptions (catches *System.Exception*) and does not deliberately throw any exception from catch blocks. <sup>5</sup>

### 7.1.4 Throw Specifier

**throw** specifier emphasizes that the function might throw a Cmajor exception.

### 7.1.5 Unit Test Specifier

**unit\_test** specifier has a meaning in programs intended as input to the **cmunit** program.

## 7.2 Function Templates

A function that contains a template parameter list is said to be a *function template*. A function template is parameterized by one or more type parameters. An optional default value can be provided for a template parameter. There cannot be a non-default template parameter coming after a default template parameter.

### 7.2.1 Constrained Function Templates

A function template that has a where-constraint (10.5) is said to be a *constrained function template*. The where-constraint sets requirements for template type parameters.

### 7.2.2 Function Template and Overload Resolution

Function templates participate in the overload resolution process. In overload resolution the template parameters are deduced from the types of function call arguments. The type parameters can also be explicitly specified using a template identifier (4.20). After the type parameters are deduced or explicitly specified, the function template is instantiated, and it becomes a concrete function. The type-checking of an unconstrained function template is deferred until instantiation time.

## 8 Derived Types

### 8.1 Pointer Types

If  $\tau$  is a type expression (4.21),  $\tau^*$  creates a new type representing a pointer to an object of type  $\tau$ . The default value of a pointer type object is **null**.

---

<sup>4</sup>name mangling encodes parameter types to the function name

<sup>5</sup>semantics changed from version 2.0.

Unary operators `*` and `-`, prefix operators `++` and `--`, binary operators `+`, `-`, `==` and `<`, and the indexing operator `[]` operate on pointer type operands.

A pointer type value can be implicitly converted to a generic pointer type **void\***.

### 8.1.1 Generic Pointer Type

The type **void\*** represents a generic pointer to a memory location or it is **null**. The memory location can hold an object or it can be a location in the free store that may or may not hold an object. The default value of a generic pointer type object is **null**.

Generic pointers can be compared for equality using the `==` operator. The generic pointer type can be converted to any other pointer type with an explicit **cast**.

### 8.1.2 Null Pointer Type

The null pointer type represents a pointer that does not point to any object or any location in the free store. The value of a null pointer type object is **null**. Any pointer type can be compared with **null** for equality using the `==` operator.

## 8.2 Lvalue Reference Types

If  $\tau$  is a type expression (4.21),  $\tau\&$  creates a new type representing an lvalue reference to an object of type  $\tau$ . Lvalue reference type objects do not have a default value and they must be initialized to refer to an object in some memory location.

If  $T$  is an object type, type  $T\&$  supports the same operators as  $T$  does.

Type  $T\&$  (an lvalue reference to an object of type  $T$ ) can be implicitly converted to type **const**  $T\&$  (an lvalue reference to a constant object of type  $T$ ).

## 8.3 Rvalue Reference Types

If  $\tau$  is a type expression (4.21),  $\tau\&\&$  creates a new type representing an rvalue reference to an object of type  $\tau$ . Rvalue references are used to implement move semantics.

## 8.4 Array Types

If  $\tau$  is a type expression (4.21),  $\tau[N]$  creates a new type representing an array whose element type is  $\tau$  and dimension is  $N$ .

Currently only single-dimensional arrays whose element type is a basic type (3) or a class type (9.2) are supported. The dimension  $N$  must be a constant expression (4.22).

Indexing operator `[]` operates on array type operands (4.14.5).

## 8.5 Constant Types

If  $\tau$  is a type expression, **const**  $\tau$  creates a new type representing a constant object of type  $\tau$ . The default value of type **const**  $\tau$  is the same as the default value of type  $\tau$ .

If  $T$  is an object type, type **const**  $T$  supports those operators of type  $T$  that do not change the value of object of type  $T$ .

Type  $T$  can be implicitly converted to type **const**  $T$ .

## 9 User-defined Types

### 9.1 Enumerated Types

An enumerated type is used to create an integral type with a set of named constants. Each constant will have a value of type **int** that is distinct from the values of other constants within the same enumerated type.

$\langle \text{enumerated-type-definition} \rangle ::= \langle \text{access-specifiers} \rangle^6 \text{enum } \langle \text{identifier} \rangle^{2.3}$   
 $\quad \text{'(' } \langle \text{enumeration-constant} \rangle (, \langle \text{enumeration-constant} \rangle)^* \text{' )'}$

$\langle \text{enumeration-constant} \rangle ::= \langle \text{identifier} \rangle (= \langle \text{constant-expression} \rangle^{4.22})?$

The default value of an enumerated type object is 0, whether it is represented by a named constant or not.

If the first constant is not explicitly given a value, it will have a value 0. If any other constant is not explicitly given a value, it will have the value of the previous constant + 1.

### 9.2 Class Types

A class type is used to define a user-defined type that can contain data members, type aliases, constants, member functions and subtypes.

$\langle \text{class-definition} \rangle ::= \langle \text{class-specifiers} \rangle \text{class } \langle \text{identifier} \rangle^{2.3} \langle \text{template-parameter-list} \rangle^{7?}$   
 $\quad \langle \text{inheritance} \rangle? \langle \text{where-constraint} \rangle^{10.5?} \text{'{' } \langle \text{class-content} \rangle \text{'}'}$

$\langle \text{class-specifiers} \rangle ::= (\langle \text{access-specifier} \rangle^6 | \text{static} | \text{abstract})^*$

$\langle \text{inheritance} \rangle ::= : (\langle \text{template-id} \rangle^{4.20} | \langle \text{qualified-id} \rangle^{2.3})$

$\langle \text{class-content} \rangle ::= (\langle \text{static-constructor} \rangle^{9.2.7}$   
|  $\langle \text{constructor} \rangle^{9.2.8}$   
|  $\langle \text{destructor} \rangle^{9.2.9}$   
|  $\langle \text{member-function} \rangle^{9.2.10}$   
|  $\langle \text{conversion-function} \rangle^{9.2.11}$   
|  $\langle \text{enumerated-type-definition} \rangle^{9.1}$   
|  $\langle \text{constant-declaration} \rangle^{9.2.13}$   
|  $\langle \text{member-variable-declaration} \rangle^{9.2.12}$   
|  $\langle \text{typedef-declaration} \rangle^{9.2.14}$   
|  $\langle \text{delegate-definition} \rangle^{9.3.1}$   
|  $\langle \text{class-delegate-definition} \rangle^{9.3.2}$   
|  $\langle \text{class-definition} \rangle)^*$

A class type can be regular, **static** or **abstract**.

#### 9.2.1 Regular Class Types

A regular class can contain each kind of class content members.

### 9.2.2 Static Class Types

A **static** class can contain

- a static constructor
- static data members
- static member functions
- constants
- typedef declarations
- delegate and class delegate definitions
- subtypes

### 9.2.3 Abstract Class Types

A class that contains one or more abstract member functions shall be declared **abstract**. One cannot create an object of an abstract class. Each concrete class that derives from an abstract class shall override each of the abstract member functions of an abstract base class.

### 9.2.4 Class Templates

A class that contains a template parameter list is said to be a *class template*. A class template is parameterized by one or more type parameters. When a template name (4.20) is specified with the type argument expressions, the class template is instantiated, and becomes a concrete class. The type-checking of a class template is deferred until instantiation time.

### 9.2.5 Inheritance

A class or a class template can derive from a class or a class template. Derived class *inherits* members of the *base class*. A derived class can *override* virtual, overridden and abstract member functions of the base class.

### 9.2.6 Constrained Class Templates

A class template that has a where-constraint (10.5) is said to be a *constrained class template*. The where-constraint sets requirements for template type parameters.

### 9.2.7 Static Constructor

The identifier of the static constructor shall be equal to the name of the class type. The purpose of the static constructor is to initialize the static member variables of a class object on its first call. The static constructor is called from each constructor and each static member function of a class type before taking other actions.

$\langle \text{static-constructor} \rangle ::= \text{static } \langle \text{static-constructor-class-name} \rangle ' ( ' ) '$   
 $( : \langle \text{static-initializer-list} \rangle ) ? \langle \text{function-body} \rangle$ <sup>7</sup>

$\langle \text{static-constructor-class-name} \rangle ::= \langle \text{identifier} \rangle$ [2.3](#)

$\langle \text{static-initializer-list} \rangle ::= \langle \text{static-initializer} \rangle (', ' \langle \text{static-initializer} \rangle)^*$

$\langle \text{static-initializer} \rangle ::= \langle \text{identifier} \rangle ' ( ' \langle \text{argument-list} \rangle$ [4.14](#)  $) '$

### 9.2.8 Constructors

The identifier of the constructor shall be equal to the name of the class type. The purpose of a constructor is to initialize non-static member variables of a class type, and to establish a class invariant.

An initializer of a constructor can delegate initialization to another constructor of the same class using the **this** keyword. The initializer can call a base class constructor using the **base** keyword.

If a non-static class has no user-defined constructor, the compiler will implement a default constructor that constructs each non-static member variable to its default value.

If a non-static class has no user-defined copy constructor, no user-defined move constructor, no user-defined copy assignment, no user-defined move assignment and no user-defined destructor, the compiler will implement a copy constructor that copy-constructs each non-static member variable from the corresponding member variable of the constructor argument. The compiler will also implement a move constructor that moves each non-static member variable from the corresponding member variable of the constructor argument in that case.

This automatic generation of default constructor, copy constructor and move constructor can be suppressed using the **suppress** keyword.

Automatic generation of default constructor, copy constructor and move constructor can be requested by using the **default** keyword.

$\langle \text{constructor} \rangle ::= \langle \text{constructor-specifiers} \rangle \langle \text{constructor-class-name} \rangle$   
 $' ( ' \langle \text{parameter-list} \rangle$ [7?](#)  $) ' ( : \langle \text{initializer-list} \rangle ) ? \langle \text{function-body} \rangle$ [7](#)

$\langle \text{constructor-specifiers} \rangle ::= ( \langle \text{access-specifier} \rangle$ [6](#)  $| \text{suppress} | \text{default} | \text{inline} | \text{nothrow} |$   
 $\text{throw} )^*$

$\langle \text{constructor-class-name} \rangle ::= \langle \text{identifier} \rangle$ [2.3](#)

$\langle \text{initializer-list} \rangle ::= \langle \text{initializer} \rangle (', ' \langle \text{initializer} \rangle)^*$

$\langle \text{initializer} \rangle ::= ( \langle \text{identifier} \rangle | \text{this} | \text{base} ) ' ( ' \langle \text{argument-list} \rangle$ [4.14](#)  $) '$

```
1 public class Foo
2 {
3     public Foo()                // default constructor
4     {
5         // ...
6     }
7     public Foo(const Foo& that) // copy constructor
8     {
9         // ...
10    }
```

```

11     public Foo(Foo&& that)          // move constructor
12     {
13         // ...
14     }
15 }
16
17 public class Bar
18 {
19     default Bar();                  // compiler will implement memberwise
        default constructor
20     default Bar(const Bar&);        // compiler will implement memberwise
        copy constructor
21     default Bar(Bar&&);              // compiler will implement memberwise
        move constructor
22 }
23
24 public class Baz
25 {
26     suppress Baz();                // automatic generation of default
        constructor is suppressed
27     suppress Baz(const Baz&);       // automatic generation of copy
        constructor is suppressed
28     suppress Baz(Baz&&);             // automatic generation of move
        constructor is suppressed
29 }

```

### 9.2.9 Destructor

The identifier of the destructor shall be equal to the name of the class type. The purpose of a destructor is to destroy non-static member variables of a class type, and to tear down the class invariant.

$\langle \text{destructor} \rangle ::= \langle \text{destructor-specifiers} \rangle \sim \langle \text{destructor-class-name} \rangle '()' \langle \text{function-body} \rangle$ <sup>7</sup>

$\langle \text{destructor-specifiers} \rangle ::= ((\langle \text{access-specifier} \rangle$ <sup>6</sup> | **virtual** | **override** | **default** | **inline**)\*

$\langle \text{destructor-class-name} \rangle ::= \langle \text{identifier} \rangle$ <sup>2.3</sup>

If the class type has a base class that has a virtual destructor, the destructor shall be declared with **override** specifier; otherwise, if the class type contains virtual functions, the destructor is automatically set **virtual**.

If a non-static class has no user-defined destructor, no user-defined copy constructor, no user-defined move constructor, no user-defined copy assignment, no user-defined move assignment, and (1) the class has virtual functions, or (2) at least one member variable has a non-trivial destructor, or (3) a class has a base class that has a non-trivial destructor, the compiler will implement a destructor that destroys all non-static member variables.

Automatic generation of destructor can be requested by using the **default** keyword.

```

1 public class Foo
2 {
3     public ~Foo()    // destructor

```

```

4      {
5          // ...
6      }
7  }
8
9  public class Bar
10 {
11     default ~Bar(); // compiler will implement memberwise destructor
12 }

```

### 9.2.10 Member Functions

A member function operates on member variables a class object or class type and can call other member functions of its class. A member function can be virtual, overridden, abstract, static or regular.

$\langle \text{member-function} \rangle ::= \langle \text{member-function-specifiers} \rangle \langle \text{type-expression} \rangle$ <sup>4.21</sup>  
 $\langle \text{function-name} \rangle$ <sup>7</sup> '('  $\langle \text{parameter-list} \rangle$ <sup>7?</sup> ')' **const**? ( $\langle \text{function-body} \rangle$ <sup>7</sup> | ;)

$\langle \text{member-function-specifiers} \rangle ::= (\langle \text{access-specifier} \rangle$ <sup>6</sup>  
 | **virtual** | **override** | **abstract** | **static** | **suppress** | **default** | **inline** | **nothrow** |  
**throw**)\*

A regular member function may have access specifiers but no other member function specifiers. A regular member function operates on a class object.

Member function can be declared *virtual*, in which case the actual member function called through a pointer or reference variable depends on the actual type of the class object instead of the formal type of the pointer or reference variable.

Member function can be declared *abstract*, in which case it shall not have an implementation, and it has to be overridden in each concrete class type derived from the abstract class type.

A member function of a derived class type can *override* a virtual or abstract member function of a base class type.

Member function can be declared *static*, in which case it do not operate on a class object (it has no **this** pointer). A static member function can call other static member functions of its class and can operate on static member variables of its class. A static member function is called using `ClassName.StaticMemberFunc(arguments)` syntax.

A member function can be declared *const*, in which case the type of the **this** pointer for a class C is "pointer to **const** C". A constant member function cannot directly alter its class object.

If a non-static class has no user-defined copy assignment, no user-defined copy constructor, no user-defined move assignment, no user-defined move constructor and no user-defined destructor, the compiler will implement a copy assignment that assigns each non-static member variable from its corresponding member variable of the function argument. The compiler will also implement a move assignment that swaps each non-static member variable with the corresponding member variable of the function argument in that case.

This automatic generation of copy assignment and move assignment can be suppressed using the **suppress** keyword.

Automatic generation of copy assignment and move assignment can be requested by using the **default** keyword.

```

1 public class Foo
2 {
3     public void operator=(const Foo& that)    // copy assignment
4     {
5         // ...
6     }
7     public void operator=(Foo&& that)        // move assignment
8     {
9         // ...
10    }
11 }
12
13 public class Bar
14 {
15     default void operator=(const Bar&); // compiler will implement
16     // memberwise copy assignment
17     default void operator=(Bar&&);      // compiler will implement
18     // memberwise move assignment
19 }
20
21 public class Baz
22 {
23     suppress void operator=(const Baz&); // automatic generation of
24     // copy assignment is suppressed
25     suppress void operator=(Baz&&);      // automatic generation of
26     // move assignment is suppressed
27 }

```

### 9.2.11 Conversion Functions

A conversion function converts a class type object to the specified target type. Conversion functions participate type conversions in cast-expressions(4.17) and in function overloading.

$\langle \text{conversion-function} \rangle ::= \langle \text{conversion-function-specifiers} \rangle \text{operator} \langle \text{type-expression} \rangle$  4.21  $\text{'('}$   
 $\text{'')' const? } (\langle \text{function-body} \rangle$  7  $| ;)$

$\langle \text{conversion-function-specifiers} \rangle ::= (\langle \text{access-specifier} \rangle$  6  $| \text{nothrow} | \text{throw} | \text{inline})^*$

### 9.2.12 Member Variable Declarations

A class can have non-static and static member variables.

$\langle \text{member-variable-declaration} \rangle ::= \langle \text{member-variable-specifiers} \rangle \langle \text{type-expression} \rangle$  4.21  $\langle \text{identifier} \rangle$  2.3  
 $;$

$\langle \text{member-variable-specifiers} \rangle ::= (\langle \text{access-specifier} \rangle$  6  $| \text{static})^*$

A non-static member variable is a member of each class object of its class.

A static member variable is a member of its class type.



### 9.2.13 Constant Declarations

A constant is an object whose value can be obtained at compile time.

$\langle \text{constant-declaration} \rangle ::= \langle \text{access-specifiers} \rangle_6$   
 $\text{const } \langle \text{type-expression} \rangle_{4.21} \langle \text{identifier} \rangle_{2.3} = \langle \text{constant-expression} \rangle_{4.22} ;$

The type of a constant can be a basic value type or an enumerated type (see 9.1).

### 9.2.14 Typedef Declarations

The **typedef** declaration inserts an alias name for a type expression into its scope.

$\langle \text{typedef-declaration} \rangle ::= \langle \text{access-specifiers} \rangle_6 \text{ typedef } \langle \text{type-expression} \rangle_{4.21} \langle \text{identifier} \rangle_{2.3} ;$

## 9.3 Delegate Types

Delegate types make possible to delegate execution of a function to another entity. Delegates can be passed as parameters and called much the same way ordinary functions can be called. Delegates are typically used to implement callbacks and events.

Delegate types come in two flavors: regular delegates that capture a function pointer and class delegates that capture a pointer to a class object and a pointer to a member function.

### 9.3.1 Delegates

A delegate type is defined using the keyword **delegate** followed by a return type, the name of the delegate type and a list of function parameters. An object of a delegate type is a pointer to a free function or to a static member function.

$\langle \text{delegate-definition} \rangle ::= (\langle \text{access-specifier} \rangle_6 | \text{throw} | \text{nothrow})^* \text{ delegate}$   
 $\langle \text{type-expression} \rangle_{4.21} \langle \text{identifier} \rangle_{2.3} \langle \text{parameters} \rangle_7 ;$

### 9.3.2 Class Delegates

A class delegate type is defined using the keyword pair **class delegate** followed by a return type, the name of the class delegate type and a list of member function parameters. An object of a class delegate type is a pair consisting of a pointer to a class object and a pointer to a member function.

$\langle \text{class-delegate-definition} \rangle ::= (\langle \text{access-specifier} \rangle_6 | \text{throw} | \text{nothrow})^* \text{ class delegate}$   
 $\langle \text{type-expression} \rangle_{4.21} \langle \text{identifier} \rangle_{2.3} \langle \text{parameters} \rangle_7 ;$

## 10 Concepts

Concepts are used to set requirements for template type arguments in function and class templates. The concept design is influenced by the Concept Design for the STL ([6]).

A concept definition consists of the name of the concept followed by a list of type parameters. A concept body consists of syntactic requirements for type parameters that take form of constraints, and semantic requirements for type parameters that take form of axioms.

A concept can *refine* another concept by setting additional requirements and overriding existing requirements. When overriding existing requirements, the refining concept can set an associated type equal to a more constrained type, or to model a more constrained concept than in the refined concept.

```

<concept-definition> ::= <access-specifiers>6
    concept <identifier>2.3 <<identifer> (, <identifier>)* >
    <refinement>? <where-constraint>10.5? { <concept-body> }

<refinement> ::= : <concept-name> <<identifer> (, <identifier>)* >

<concept-name> ::= <qualified-id>2.3

<concept-body> ::= (<typename-constraint>10.1
    | <signature-constraint>10.2
    | <embedded-constraint>10.3
    | <axiom>10.4)*

```

## 10.1 Typename Constraint

A typename constraint sets a requirement that an associated type with the specified name is found.

```

<typename-constraint> ::= typename <type-expression>4.21 ;

```

Typically type expression is of the form *<concept type parameter> . <associated type name>*.

## 10.2 Signature Constraint

A signature constraint sets a requirement that either the first concept type parameter contains a member function with the specified signature or there exists a nonmember function with the specified signature. The signature of the function does not have to match exactly.

```

<signature-constraint> ::= <constructor-constraint>
    | <destructor-constraint>
    | <member-function-constraint>
    | <function-constraint>

<constructor-constraint> ::= <identifier>2.3 <parameters>7 ;

<destructor-constraint> ::= ~ <identifier> '(' ' ' ' ' ;

<member-function-constraint> ::= <type-expression>4.21
    <identifier> . <function-name>7 <parameters> ;

<function-constraint> ::= <type-expression>4.21 <function-name>7 <parameters> ;

```

### 10.3 Embedded Constraint

An embedded constraint sets additional requirements for concept parameter types.

$\langle \text{embedded-constraint} \rangle ::= \langle \text{where-constraint} \rangle$ [10.5](#) ;

### 10.4 Axioms

A concept can contain *axioms* that represent semantic requirements for constrained template type arguments. The axioms are not processed in any way by the compiler (except parsing their syntax). They are only documentation for the programmer.

$\langle \text{axiom} \rangle ::= \text{axiom } \langle \text{identifier} \rangle$ [2.3](#)  $\langle \text{parameters} \rangle$ [7](#) {  $\langle \text{axiom-body} \rangle$  }

$\langle \text{axiom-body} \rangle ::= \langle \text{axiom-statement} \rangle^*$

$\langle \text{axiom-statement} \rangle ::= \langle \text{expression} \rangle$ [4](#) ;

### 10.5 Where Constraint

A where constraint sets requirements for template type arguments or concept parameter types.

$\langle \text{where-constraint} \rangle ::= \text{where } \langle \text{constraint-expr} \rangle$

$\langle \text{constraint-expr} \rangle ::= \langle \text{disjunctive-constraint-expr} \rangle$ [10.5.1](#)

#### 10.5.1 Disjunctive Constraint Expressions

A disjunctive constraint expression states that the template type arguments and concept parameters must satisfy the requirements of one of the conjunctive constraint expressions separated by **or** connectives.

$\langle \text{disjunctive-constraint-expr} \rangle ::= \langle \text{conjunctive-constraint-expr} \rangle$ [10.5.2](#) (**or**  $\langle \text{conjunctive-constraint-expr} \rangle$ )\*

#### 10.5.2 Conjunctive Constraint Expressions

A conjunctive constraint expression states that the template type arguments and concept parameters must satisfy the requirements of all primary constraint expressions separated by **and** connectives.

$\langle \text{conjunctive-constraint-expr} \rangle ::= \langle \text{primary-constraint-expr} \rangle$ [10.5.3](#) (**and**  $\langle \text{primary-constraint-expr} \rangle$ )\*

#### 10.5.3 Primary Constraint Expressions

A primary constraint can be either an atomic constraint or a constraint expression enclosed in parenthesis.

$\langle \text{primary-constraint-expr} \rangle ::= \langle \text{atomic-constraint} \rangle$ [10.5.4](#) | '('  $\langle \text{constraint-expr} \rangle$ [10.5](#) ')'

#### 10.5.4 Atomic Constraints

An atomic constraint can be either an is-constraint or a multiparameter constraint.

$\langle atomic\_constraint \rangle ::= \langle is\_constraint \rangle 10.5.5 \mid \langle multiparam\_constraint \rangle 10.5.6$

#### 10.5.5 Is-Constraint

An is-constraint sets a requirement that a either (1) the specified type satisfies the requirements of the specified concept, or (2) the specified type is trivially convertible to another type. For example, type **const T&** is trivially convertible to type **T**. When specifying the is-constraint in case (2), the so far less constrained type should be put to the left and the more constrained type to the right.

$\langle is\_constraint \rangle ::= \langle type\_expression \rangle 4.21 \text{ is } \langle concept\_or\_typename \rangle$

$\langle concept\_or\_typename \rangle ::= \langle type\_expression \rangle$

#### 10.5.6 Multiparameter Constraint

A multiparameter constraint sets a requirement that the specified types satisfy the requirements of the specified concept.

$\langle multiparam\_constraint \rangle ::= \langle concept\_name \rangle 10 < \langle type\_expression \rangle 4.21 (, \langle type\_expression \rangle)^* >$

### 10.6 Built-in Concepts

There are four built-in multiparameter concepts implemented by the compiler.

#### 10.6.1 Same Concept

The **Same**<**T**, **U**> concept sets a requirement that types **T** and **U** are exactly the same type. The so far less constrained type should be put to the left and the more constrained type to the right.

#### 10.6.2 Derived Concept

The **Derived**<**T**, **U**> concept sets a requirement that type **T** is derived from type **U**, or in other words that the type **U** is the base class of type **T**. Type **T** is trivially derived from type **T**.

#### 10.6.3 Convertible Concept

The **Convertible**<**T**, **U**> concept sets a requirement that type **T** is convertible to type **U**. Either there should be a constructor with signature **U(const T&)** or there should be a built-in conversion from type **T** to type **U**. Type **T** is trivially convertible to type **T**.

#### 10.6.4 Common Concept

The **Common**<T, U> concept sets a requirement that types T and U have a common type to which they both are convertible to. The common type is inserted as a typedef *CommonType* to the scope of concept **Common**<T, U>. The common type of T and T is trivially T. The so far less constrained type should be put to the left and the more constrained type to the right.

## 11 Namespaces

Namespaces are used to disambiguate equal names belonging to different libraries. A namespace consists of optional using directives followed by optional definitions.

$\langle \text{namespace-definition} \rangle ::= \text{namespace } \langle \text{qualified-id} \rangle_{2.3} \{ \langle \text{namespace-content} \rangle \}$

$\langle \text{namespace-content} \rangle ::= \langle \text{using-directive} \rangle^* \langle \text{definition} \rangle^*$

$\langle \text{using-directive} \rangle ::= \langle \text{using-alias-directive} \rangle \mid \langle \text{using-namespace-directive} \rangle$

$\langle \text{using-alias-directive} \rangle ::= \text{using } \langle \text{identifier} \rangle_{2.3} = \langle \text{qualified-id} \rangle ;$

$\langle \text{using-namespace-directive} \rangle ::= \text{using } \langle \text{qualified-id} \rangle ;$

$\langle \text{definition} \rangle ::= \langle \text{enumerated-type-definition} \rangle_{9.1}$

- |  $\langle \text{constant-declaration} \rangle_{9.2.13}$
- |  $\langle \text{function-definition} \rangle_7$
- |  $\langle \text{class-definition} \rangle_{9.2}$
- |  $\langle \text{delegate-definition} \rangle_{9.3.1}$
- |  $\langle \text{class-delegate-definition} \rangle_{9.3.2}$
- |  $\langle \text{typedef-declaration} \rangle_{9.2.14}$
- |  $\langle \text{concept-definition} \rangle_{10}$
- |  $\langle \text{namespace-definition} \rangle$

A using alias directive brings a single alias name to the current file scope.

A using namespace directive brings contents of given namespace to the current file scope.

## 12 Source Files

A source file consists of namespace content.

$\langle \text{source-file} \rangle ::= \langle \text{namespace-content} \rangle_{11}$

## 13 Programs

A Cmajor program consists of source files. One of the source files must contain a main function.

A main function can have four possible signatures:

- `void main()`
- `int main()`
- `void main(int argc, const char** argv)`
- `int main(int argc, const char** argv)`

In the two latter signatures parameter `argc` contains the number of program arguments, and parameter `argv` contains program argument strings. By convention the first argument (`argv[0]`) contains the name of the program.

If the main function is declared to return a value, the main function shall contain a return statement, otherwise it shall not contain a return statement.

If the main function is declared void, it returns exit code 0 to the environment.

## 14 Solution and Project File Formats

### 14.1 Solution File Format

$\langle \text{solution-file} \rangle ::= \langle \text{solution-declaration} \rangle \langle \text{project-file-declarations} \rangle \langle \text{project-dependencies} \rangle$

$\langle \text{solution-declaration} \rangle ::= \text{solution } \langle \text{solution-name} \rangle \text{ ';'}$

$\langle \text{solution-name} \rangle ::= \langle \text{qualified-id} \rangle$  [2.3](#)

$\langle \text{project-file-declarations} \rangle ::= \langle \text{project-file-declaration} \rangle^*$

$\langle \text{project-file-declaration} \rangle ::= \text{project } \langle \text{project-file-path} \rangle \text{ ';'}$

$\langle \text{project-file-path} \rangle ::= \text{'<' } [\text{^>}]^+ \text{'>'}$

$\langle \text{project-dependencies} \rangle ::= \langle \text{project-dependency} \rangle^*$

$\langle \text{project-dependency} \rangle ::= \text{dependency } \langle \text{project-name} \rangle \langle \text{dependent-project-list} \rangle \text{ ';'}$

$\langle \text{dependent-project-list} \rangle ::= \text{'(' } \langle \text{dependent-project-name} \rangle \text{' ,' } \langle \text{dependent-project-name} \rangle^* \text{' )'}$

$\langle \text{dependent-project-name} \rangle ::= \langle \text{qualified-id} \rangle$

### 14.2 Project File Format

$\langle \text{project-file} \rangle ::= \langle \text{project-declaration} \rangle \langle \text{project-file-declarations} \rangle$

$\langle \text{project-declaration} \rangle ::= \text{project } \langle \text{project-name} \rangle \text{ ';'}$

$\langle \text{project-name} \rangle ::= \langle \text{qualified-id} \rangle$  [2.3](#)

$\langle \text{project-file-declarations} \rangle ::= \langle \text{project-file-declaration} \rangle^*$

$\langle \text{project-file-declaration} \rangle ::= \langle \text{target-declaration} \rangle$   
 $\quad | \langle \text{assembly-declaration} \rangle$   
 $\quad | \langle \text{library-reference} \rangle$   
 $\quad | \langle \text{executable-declaration} \rangle$   
 $\quad | \langle \text{c-library-declaration} \rangle$   
 $\quad | \langle \text{source-file} \rangle$   
 $\quad | \langle \text{asm-source-file} \rangle$   
 $\quad | \langle \text{c-source-file} \rangle$   
 $\quad | \langle \text{text-file} \rangle$

$\langle \text{target-declaration} \rangle ::= \text{target '}' (\text{program} | \text{library}) ';$

$\langle \text{assembly-declaration} \rangle ::= \text{assembly} \langle \text{file-path} \rangle \langle \text{properties} \rangle? ';$

$\langle \text{library-reference} \rangle ::= \text{reference} \langle \text{file-path} \rangle \langle \text{properties} \rangle? ';$

$\langle \text{executable-declaration} \rangle ::= \text{executable} \langle \text{file-path} \rangle ';$

$\langle \text{c-library-declaration} \rangle ::= \text{clib} \langle \text{file-path} \rangle \langle \text{properties} \rangle? ';$

$\langle \text{source-file} \rangle ::= \text{source} \langle \text{file-path} \rangle \langle \text{properties} \rangle? ';$

$\langle \text{asm-source-file} \rangle ::= \text{asmsource} \langle \text{file-path} \rangle \langle \text{properties} \rangle? ';$

$\langle \text{c-source-file} \rangle ::= \text{csource} \langle \text{file-path} \rangle \langle \text{properties} \rangle? ';$

$\langle \text{text-file} \rangle ::= \text{text} \langle \text{file-path} \rangle \langle \text{properties} \rangle? ';$

$\langle \text{properties} \rangle ::= '[' \langle \text{property} \rangle (',' \langle \text{property} \rangle)^* ']$

$\langle \text{property} \rangle ::= \langle \text{property-name} \rangle '=' \langle \text{property-value} \rangle$

$\langle \text{property-name} \rangle ::= \langle \text{identifier} \rangle$

$\langle \text{property-value} \rangle ::= \langle \text{identifier} \rangle$

$\langle \text{file-path} \rangle ::= '<' [\wedge]+ '>'$

Properties can be used to include a declaration only in selected configurations. Currently property name can be **backend** or **os**. Backend values can be **c** and **llvm**. Os values can be **windows** and **linux**.

## References

- [1] AHO, A. V., M. S. LAM, R. SETHI, AND J. D. ULLMAN: Compilers: Principles, Techniques, & Tools. Second Edition. Addison-Wesley, 2007.
- [2] Boost C++ libraries, <http://www.boost.org/>
- [3] ELLIS, M. A., AND B. STROUSTRUP: The Annotated C++ Reference Manual. Addison-Wesley, 1990.
- [4] GIBBS, M., AND B. STROUSTRUP: Fast dynamic casting, 2005, [http://www.stroustrup.com/fast\\_dynamic\\_casting.pdf](http://www.stroustrup.com/fast_dynamic_casting.pdf)
- [5] HAN THE THANH: pdfTEX, <http://www.tug.org/applications/pdftex/>
- [6] JTC1/SC22/WG21 - THE C++ STANDARDS COMMITTEE: A Concept Design for the STL, <http://www.open-std.org/jtc1/sc22/wg21/docs/papers/2012/n3351.pdf>
- [7] JTC1/SC22/WG21 - THE C++ STANDARDS COMMITTEE: Working Draft, Standard for Programming Language C++, <http://www.open-std.org/jtc1/sc22/wg21/docs/papers/2011/n3242.pdf>
- [8] LAAKKO, S.: Soul Parsing Framework, <http://sourceforge.net/projects/soulparsing/>
- [9] MICROSOFT CORPORATION: The C# Language Specification. Version 4.0, <http://go.microsoft.com/fwlink/?LinkId=199552>
- [10] PATTERSON, D. A., AND J. L. HENNESSY: Computer Organization and Design. The Hardware / Software Interface. Fourth Edition. Morgan Kaufmann, 2012.
- [11] STEPANOV, A., AND P. MCJONES: Elements of Programming. Addison-Wesley, 2009.
- [12] STROUSTRUP, B.: The C++ Programming Language. Fourth Edition. Addison-Wesley, 2013.
- [13] SUNDARESAN V., AND C. RAZAFIMAHEFA, R. VALLEE-RAI, L. HENDREN, P. LAM, E. CAGNON, C. GODIN: Practical Virtual Method Call Resolution for Java, 1999, <http://web.cs.ucla.edu/~palsberg/tba/papers/sundaresan-et-al-oopsla00.pdf>